Differentiation -- Finite Difference Methods

In[657]:=
```
x = 0; Remove["Global`*"]; DateList[Date[]] // Most
```

Out[657]= {2020, 3, 17, 21, 52}

This notebook derives various finite difference approximations for derivatives. It applies them to the function used in Exercise 4, Chapter 2.

In[658]:=

$$\text{ftest}[x\_] = \frac{4970\,x - 4923}{4970\,x^2 - 9799\,x + 4830};$$

```
fp = ftest'[1]; fpp = ftest''[1];
{ftest[1], fp, fpp, ftest'''[1]}
```

Out[660]= {47, -1657, 94, 49 371 978}

One-sided formula for first derivative

Define a function which represents the one-sided finite difference formula for $f'(x)$.

In[661]:=
```
fdif[x_, eps_] := (ftest[eps + x] - ftest[x]) / eps
```

List the finite difference function for various values of eps. Below is a table presenting results for various choices of eps. We execute fdif twice. First, we set all the arguments equal to rational numbers, which causes *Mathematica* to use infinite precision arithmetic, and report the real value. The column with the "inf. prec." heading gives the estimated derivative, and the next column with heading "error" gives the relative error of this infinite precision result. These columns indicate the errors that would occur if we had infinite precision. Unfortunately, we generally don't. Therefore, we have to be careful about our choice for eps. Second, we use floating point arithmetic. The second case corresponds to what a computer would typically do. The "finite precision" column gives us the resulting finite precision result and the last column reports the relative error of the finite precision result.

In[662]:=
```
headings = {"epsilon", "inf. prec.", "error", "finite prec.", " error"};
Results := Block[{h, i, exactdif, finprecdif, x0f, hf},
   Table[
    (*Specify finite difference, both infinite precision and finite precision*)
    h = 10^-i; hf = N[h];
    (* Set finite precision value of x0*)
    x0f = N[x0];
    (*Compute the exact finite difference and the machine precision finite difference*)
    exactdif = fdif[x0, h]; finprecdif = fdif[x0f, hf];
    (*Output the finite differnce,
    the infinite precision finite difference, its error, finite precision, and error*)
    {N[h],
     exactdif // N, ScientificForm[N[exactdif / truth] - 1, 3],
     finprecdif , ScientificForm[finprecdif / truth - 1, 3]},
    {i, 2, 15}]];
```

In[664]:=
```
truth = fp; x0 = 1;
TableForm[Join[{headings}, Results]]
```

Out[665]//TableForm=

epsilon	inf. prec.	error	finite prec.	error
0.01	-1373.55	-1.71×10^{-1}	-1373.55	-1.71×10^{-1}
0.001	-1649.77	-4.36×10^{-3}	-1649.77	-4.36×10^{-3}
0.0001	-1656.91	-5.18×10^{-5}	-1656.91	-5.18×10^{-5}
0.00001	-1657.	-7.8×10^{-7}	-1657.	-7.78×10^{-7}
$1.\times10^{-6}$	-1657.	-3.33×10^{-8}	-1657.	-3.56×10^{-8}
$1.\times10^{-7}$	-1657.	-2.89×10^{-9}	-1657.	1.01×10^{-7}
$1.\times10^{-8}$	-1657.	-2.84×10^{-10}	-1657.	-2.43×10^{-6}
$1.\times10^{-9}$	-1657.	-2.84×10^{-11}	-1656.95	-2.97×10^{-5}
$1.\times10^{-10}$	-1657.	-2.84×10^{-12}	-1656.95	-2.79×10^{-5}
$1.\times10^{-11}$	-1657.	-2.84×10^{-13}	-1651.37	-3.4×10^{-3}
$1.\times10^{-12}$	-1657.	-2.84×10^{-14}	-1612.53	-2.68×10^{-2}
$1.\times10^{-13}$	-1657.	-2.78×10^{-15}	-1873.56	1.31×10^{-1}
$1.\times10^{-14}$	-1657.	-2.22×10^{-16}	727.596	-1.44
$1.\times10^{-15}$	-1657.	0.	5456.97	-4.29

The infinite precision version is excellent for all h less than 10^{-6}

The floating point version is best for epsilon equal to 10^{-6}, and very good for epsilon between $10^{-6}$and 10^{-10} but smaller step sizes leads to less accurate answers.

Three-Point Formulas

Three-Point Formula for first Derivative

We first compute the general three-point formula for f'[x0]. We begin by computing the quadratic Taylor series approximation about x.

In[666]:=
```
Clear[x, x0]
Tayf[x_] = Normal[Series[f[x], {x, x0, 3}]]
```

Out[667]= $f[x0] + (x - x0)\, f'[x0] + \frac{1}{2} (x - x0)^2 f''[x0] + \frac{1}{6} (x - x0)^3 f^{(3)}[x0]$

We want to approximate f'[x0] with a finite sum of values of f[x] evaluated at x0 and two other points, x1 and x2. The weights a, b, and c are to be chosen so that a weighted sum of f[x0], f[x1], and f[x2] is a good approximation to f'[x0]. Hence we define ErrorDv to be the error in the approximation. We use the values of Tayf[x] to define ErrorDv :

In[668]:=
```
ErrorDv = Collect[
      Simplify[a Tayf[x0] + b Tayf[x1] + c Tayf[x2] - f'[x0]],
      {f[x0],f'[x0],f''[x0]}]
```

Out[668]= $(a + b + c)\, f[x0] + (-1 + b\, (-x0 + x1) + c\, (-x0 + x2))\, f'[x0] + \left(\frac{1}{2} b\, (x0 - x1)^2 + \frac{1}{2} c\, (x0 - x2)^2\right) f''[x0] + \frac{1}{6} b\, (-x0 + x1)^3 f^{(3)}[x0] + \frac{1}{6} c\, (-x0 + x2)^3 f^{(3)}[x0]$

We next solve for the values of a, b, and c which makes ErrorDv = 0 up to an error cubic in the differences x1-x0 and x2-x0. We define pt3 to be the solution:

In[669]:=
```
pt3 = Simplify[Solve[
{Coefficient[ErrorDv,f[x0]]==0,Coefficient[ErrorDv,f'[x0]]==0,Coefficient[ErrorDv,f''[x0]]==0},
{a,b,c}
]]
```

Out[669]= $\left\{\left\{a \to -\frac{-2\, x0 + x1 + x2}{(x0 - x1)\, (x0 - x2)},\ b \to \frac{-x0 + x2}{(x0 - x1)\, (x1 - x2)},\ c \to \frac{-x0 + x1}{(x0 - x2)\, (-x1 + x2)}\right\}\right\}$

The solution above is for arbitrary x0, x1, and x2. The usual three points we use are x0 and two points of equal distance to either side. We next derive the weights for the symmetric three-point formula.

In[670]:=
```
symmcase=pt3/.{x1->x0-h,x2->x0+h}
```

Out[670]= $\left\{\left\{a \to 0,\ b \to -\frac{1}{2\,h},\ c \to \frac{1}{2\,h}\right\}\right\}$

In[671]:=
```
symmrule = a f[x0] + b f[x1] + c f[x2] /. pt3[[1]] /. {x1 → x0 - h, x2 → x0 + h}
```

Out[671]= $-\frac{f[-h+x0]}{2\,h}+\frac{f[h+x0]}{2\,h}$

In[672]:=
```
Clear[fdif]
fdif[x0_, h_] = symmrule /. {f → ftest}
```

Out[673]= $-\frac{-4923+4970\,(-h+x0)}{2\,h\,\left(4830-9799\,(-h+x0)+4970\,(-h+x0)^2\right)}+\frac{-4923+4970\,(h+x0)}{2\,h\,\left(4830-9799\,(h+x0)+4970\,(h+x0)^2\right)}$

In[674]:=
```
x0 = 1;
TableForm[Join[{headings}, Results]]
```

Out[675]//TableForm=

epsilon	inf. prec.	error	finite prec.	error
0.01	3214.95	−2.94	3214.95	−2.94
0.001	−1648.65	-5.04×10^{-3}	−1648.65	-5.04×10^{-3}
0.0001	−1656.92	-4.97×10^{-5}	−1656.92	-4.97×10^{-5}
0.00001	−1657.	-4.97×10^{-7}	−1657.	-4.97×10^{-7}
$1.\times10^{-6}$	−1657.	-4.97×10^{-9}	−1657.	7.54×10^{-9}
$1.\times10^{-7}$	−1657.	-4.97×10^{-11}	−1657.	1.25×10^{-8}
$1.\times10^{-8}$	−1657.	-4.97×10^{-13}	−1657.	2.65×10^{-7}
$1.\times10^{-9}$	−1657.	-5.11×10^{-15}	−1656.97	-1.66×10^{-5}
$1.\times10^{-10}$	−1657.	0.	−1656.95	-2.78×10^{-5}
$1.\times10^{-11}$	−1657.	0.	−1651.37	-3.4×10^{-3}
$1.\times10^{-12}$	−1657.	0.	−1634.36	-1.37×10^{-2}
$1.\times10^{-13}$	−1657.	0.	−1655.28	-1.04×10^{-3}
$1.\times10^{-14}$	−1657.	0.	728.	−1.44
$1.\times10^{-15}$	−1657.	0.	−16 372.	8.88

Again, 10^{-6} is best.

In[676]:=
```
error = (ErrorDv /. pt3[[1]] /. {x1 → x0 - h, x2 → x0 + h}) + ϵ / h
```

Out[676]= $\frac{\epsilon}{h} + \frac{1}{6} h^2 f^{(3)}[1]$

In[677]:=
```
Solve[D[error, h] == 0, h]
```

Out[677]= $\left\{\left\{h \to -\frac{(-3)^{1/3} \epsilon^{1/3}}{f^{(3)}[1]^{1/3}}\right\}, \left\{h \to \frac{3^{1/3} \epsilon^{1/3}}{f^{(3)}[1]^{1/3}}\right\}, \left\{h \to \frac{(-1)^{2/3} 3^{1/3} \epsilon^{1/3}}{f^{(3)}[1]^{1/3}}\right\}\right\}$

Three-Point Formula for Second Derivative

We first compute the general three-point formula for f''[x0]. We begin by computing the quadratic Taylor series approximation about x.

In[678]:=
```
Clear[x, x0]
Tayf[x_] = Normal[Series[f[x], {x, x0, 4}]]
```

Out[679]=
$$f[x0] + (x - x0)\, f'[x0] + \frac{1}{2}(x - x0)^2 f''[x0] + \frac{1}{6}(x - x0)^3 f^{(3)}[x0] + \frac{1}{24}(x - x0)^4 f^{(4)}[x0]$$

We want to approximate f'[x0] with a finite sum of values of f[x] evaluated at x0 and two other points, x1 and x2. The weights a, b, and c are to be chosen so that a weighted sum of f[x0], f[x1], and f[x2] is a good approximation to f'[x0]. Hence we define ErrorDv to be the error in the approximation. We use the values of Tayf[x] to define ErrorDv :

In[680]:=
```
ErrorDv = Collect[
      Simplify[a Tayf[x0] + b Tayf[x1] + c Tayf[x2] - f''[x0]],
      {f[x0],f'[x0],f''[x0]}]
```

Out[680]=
$$(a + b + c)\, f[x0] + (b\,(-x0 + x1) + c\,(-x0 + x2))\, f'[x0] + \left(-1 + \frac{1}{2} b\,(x0 - x1)^2 + \frac{1}{2} c\,(x0 - x2)^2\right) f''[x0] + \frac{1}{6} b\,(-x0 + x1)^3 f^{(3)}[x0] + \frac{1}{6} c\,(-x0 + x2)^3 f^{(3)}[x0] + \frac{1}{24} b\,(x0 - x1)^4 f^{(4)}[x0] + \frac{1}{24} c\,(x0 - x2)^4 f^{(4)}[x0]$$

We next solve for the values of a, b, and c which makes ErrorDv = 0 up to an error cubic in the differences x1-x0 and x2-x0. We define pt3 to be the solution:

In[681]:=
```
pt3 = Simplify[Solve[{Coefficient[ErrorDv,f'[x0]]==0,
                Coefficient[ErrorDv,f[x0]]==0,
                Coefficient[ErrorDv,f''[x0]]==0},
              {a,b,c}]]
```

Out[681]=
$$\left\{\left\{a \to \frac{2}{(x0 - x1)\,(x0 - x2)},\ b \to -\frac{2}{(x0 - x1)\,(x1 - x2)},\ c \to -\frac{2}{(x0 - x2)\,(-x1 + x2)}\right\}\right\}$$

The solution above is for arbitrary x0, x1, and x2. The usual three points we use are x0 and two points of equal distance to either side. We next derive the weights for the symmetric three-point formula.

In[682]:=
```
symmcase=pt3/.{x1->x0-h,x2->x0+h}
```

Out[682]= $\left\{\left\{a \to -\frac{2}{h^2},\ b \to \frac{1}{h^2},\ c \to \frac{1}{h^2}\right\}\right\}$

In[683]:=
```
symmrule = a f[x0] + b f[x1] + c f[x2] /. pt3[[1]] /. {x1 → x0 - h , x2 → x0 + h }
```

Out[683]= $-\frac{2\, f[x0]}{h^2} + \frac{f[-h + x0]}{h^2} + \frac{f[h + x0]}{h^2}$

In[684]:=
```
Clear[fdif]
fdif[x0_, h_] = symmrule /. {f → ftest};
```

We now present a table of results concerning the accuracy of alternative epsilons

In[686]:=
```
x0 = 1; truth = ftest''[1];
TableForm[Join[{headings}, Results]]
```

Out[687]//TableForm=

epsilon	inf. prec.	error	finite prec.	error
0.01	-917 699.	-9.76×10^{3}	-917 699.	-9.76×10^{3}
0.001	-2250.2	-2.49×10^{1}	-2250.2	-2.49×10^{1}
0.0001	70.7882	-2.47×10^{-1}	70.7872	-2.47×10^{-1}
0.00001	93.7679	-2.47×10^{-3}	93.1603	-8.93×10^{-3}
$1.\times10^{-6}$	93.9977	-2.47×10^{-5}	142.977	5.21×10^{-1}
$1.\times10^{-7}$	94.	-2.47×10^{-7}	-2935.	-3.22×10^{1}
$1.\times10^{-8}$	94.	-2.47×10^{-9}	894 720.	9.52×10^{3}
$1.\times10^{-9}$	94.	-2.47×10^{-11}	4.32292×10^{7}	4.6×10^{5}
$1.\times10^{-10}$	94.	-2.47×10^{-13}	0.	-1.
$1.\times10^{-11}$	94.	-2.44×10^{-15}	0.	-1.
$1.\times10^{-12}$	94.	0.	4.36455×10^{13}	4.64×10^{11}
$1.\times10^{-13}$	94.	0.	-4.36561×10^{15}	-4.64×10^{13}
$1.\times10^{-14}$	94.	0.	7.03687×10^{13}	7.49×10^{11}
$1.\times10^{-15}$	94.	0.	4.36579×10^{19}	4.64×10^{17}

Let ϵ denote noise.

In[688]:=
```
error = (ErrorDv /. pt3[[1]] /. {x1 → x0 - h f[x0], x2 → x0 + h f[x0]}) + ϵ / h
```

Out[688]=
$$\frac{\epsilon}{h} + \frac{1}{12} h^2 f[1]^2 f^{(4)}[1]$$

In[689]:=
```
Solve[D[error, h] == 0, h]
```

Out[689]=
$$\left\{\left\{h \to -\frac{(-6)^{1/3} \epsilon^{1/3}}{f[1]^{2/3} f^{(4)}[1]^{1/3}}\right\}, \left\{h \to \frac{6^{1/3} \epsilon^{1/3}}{f[1]^{2/3} f^{(4)}[1]^{1/3}}\right\}, \left\{h \to \frac{(-1)^{2/3} 6^{1/3} \epsilon^{1/3}}{f[1]^{2/3} f^{(4)}[1]^{1/3}}\right\}\right\}$$

The best ϵ is now about 10^{-5}, as indicated by the table and by the theoretical result. The final error is approximated by

In[690]:=
```
error /. %[[2]]
```

Out[690]=
$$\frac{3^{2/3} \epsilon^{2/3} f[1]^{2/3} f^{(4)}[1]^{1/3}}{2 \times 2^{1/3}}$$

Four-Point Formula

First Derivative

We next compute a four-point formula, proceeding in the same fashion as with the three-point formula. We compute the cubic expansion of f[x] at x0.

In[691]:=
```
Tayf[x_] = Series[f[x],{x,x0,4}]//Normal
```

Out[691]= $$f[1]+(-1+x)\,f'[1]+\frac{1}{2}(-1+x)^2 f''[1]+\frac{1}{6}(-1+x)^3 f^{(3)}[1]+\frac{1}{24}(-1+x)^4 f^{(4)}[1]$$

We express the error between a four-point combination and f'[x0], where we approximate f at the xi with the cubic Taylor expansion.

In[692]:=
```
ErrorDv = a Tayf[x0] + b Tayf[x1] +
     c Tayf[x2] + d Tayf[x3] - f'[x0]
```

Out[692]= $$a\,f[1]-f'[1]+b\left(f[1]+(-1+x1)\,f'[1]+\frac{1}{2}(-1+x1)^2 f''[1]+\frac{1}{6}(-1+x1)^3 f^{(3)}[1]+\frac{1}{24}(-1+x1)^4 f^{(4)}[1]\right)+$$
$$c\left(f[1]+(-1+x2)\,f'[1]+\frac{1}{2}(-1+x2)^2 f''[1]+\frac{1}{6}(-1+x2)^3 f^{(3)}[1]+\frac{1}{24}(-1+x2)^4 f^{(4)}[1]\right)+$$
$$d\left(f[1]+(-1+x3)\,f'[1]+\frac{1}{2}(-1+x3)^2 f''[1]+\frac{1}{6}(-1+x3)^3 f^{(3)}[1]+\frac{1}{24}(-1+x3)^4 f^{(4)}[1]\right)$$

We find the coefficients {a,b,c,d} such that the finite difference formula approximates f[x0] and the first three derivatives correctly???

In[693]:=
```
pt4 = Simplify[Solve[{Coefficient[ErrorDv,f'[x0]]==0,
                Coefficient[ErrorDv,f[x0]]==0,
                Coefficient[ErrorDv,f''[x0]]==0,
                Coefficient[ErrorDv,f'''[x0]]==0},
              {a,b,c,d}]]
```

Out[693]= $$\left\{\left\{a\to-\frac{3+x2\,(-2+x3)-2\,x3+x1\,(-2+x2+x3)}{(-1+x1)\,(-1+x2)\,(-1+x3)},\ b\to\frac{(-1+x2)\,(-1+x3)}{(-1+x1)\,(x1-x2)\,(x1-x3)},\right.\right.$$
$$\left.\left.c\to\frac{(-1+x1)\,(-1+x3)}{(-1+x2)\,(-x1+x2)\,(x2-x3)},\ d\to-\frac{(-1+x1)\,(-1+x2)}{(x1-x3)\,(-1+x3)\,(-x2+x3)}\right\}\right\}$$

The solution pt4 assumes an arbitrary set of four points. We next compute the four-point formula where x1 is to the left of x0, and x2 and x3 are on the right:

```
In[694]:= Simplify[pt4/.{x1->x0-h,x2->x0+h,x3->x0+2 h}]
```

Out[694]= $\left\{\left\{a \to -\frac{1}{2\,h},\ b \to -\frac{1}{3\,h},\ c \to \frac{1}{h},\ d \to -\frac{1}{6\,h}\right\}\right\}$

```
In[695]:= symm = pt4[[1]]/.{x1->x0-h,x2->x0+h,x3->x0+2 h}//Simplify
```

Out[695]= $\left\{a \to -\frac{1}{2\,h},\ b \to -\frac{1}{3\,h},\ c \to \frac{1}{h},\ d \to -\frac{1}{6\,h}\right\}$

```
In[696]:= symmrule = a f[x0] + b f[x1] + c f[x2] + d f[x3] /. symm /. {x1 → x0 - h, x2 → x0 + h, x3 → x0 + 2 h} // Simplify
```

Out[696]= $-\frac{3\,f[1] + 2\,f[1-h] - 6\,f[1+h] + f[1+2\,h]}{6\,h}$

```
In[697]:= error = ( ErrorDv /. {x1 → x0 - h, x2 → x0 + h, x3 → x0 + 2 h} /. symm // Simplify) + ϵ / h
```

Out[697]= $\frac{\epsilon}{h} - \frac{1}{12}\,h^3\,f^{(4)}[1]$

In[698]:= **maxerror =** $\frac{\epsilon}{h} + \frac{1}{12}\,h^3\,M$

Out[698]= $\frac{h^3\,M}{12} + \frac{\epsilon}{h}$

```
In[699]:= Solve[D[maxerror, h] == 0, h]
```

Out[699]= $\left\{\left\{h \to -\frac{\sqrt{2}\,\epsilon^{1/4}}{M^{1/4}}\right\},\ \left\{h \to -\frac{i\,\sqrt{2}\,\epsilon^{1/4}}{M^{1/4}}\right\},\ \left\{h \to \frac{i\,\sqrt{2}\,\epsilon^{1/4}}{M^{1/4}}\right\},\ \left\{h \to \frac{\sqrt{2}\,\epsilon^{1/4}}{M^{1/4}}\right\}\right\}$

Five-Point Formula

Five-Point Formula for first derivative

We next compute the five-point formula. First, a degree 4 expansion

In[700]:=
```
Clear[x0,x,f]
Tayf[x_] = Series[f[x],{x,x0,5}]//Normal
```

Out[701]= $f[x0] + (x - x0)\, f'[x0] + \frac{1}{2}(x - x0)^2 f''[x0] + \frac{1}{6}(x - x0)^3 f^{(3)}[x0] + \frac{1}{24}(x - x0)^4 f^{(4)}[x0] + \frac{1}{120}(x - x0)^5 f^{(5)}[x0]$

Define the error expression.

In[702]:=
```
ErrorDv = a Tayf[x0] + b Tayf[x1] + c Tayf[x2] + d Tayf[x3] +
     e Tayf[x4] - f'[x0];
```

Find {a,b,c,d,e} such that the error is zero up to the fourth order.

In[703]:=
```
pt5 = Solve[
   {Coefficient[ErrorDv, f'[x0]] == 0, Coefficient[ErrorDv, f[x0]] == 0,
    Coefficient[ErrorDv, f''[x0]] == 0, Coefficient[ErrorDv, f'''[x0]] == 0, Coefficient[ErrorDv, f''''[x0]] == 0},
   {a, b, c, d, e}] // Simplify
```

Out[703]= $\Big\{\Big\{a \to -\frac{-4\,x0^3 + x2\,x3\,x4 + 3\,x0^2\,(x1 + x2 + x3 + x4) + x1\,(x3\,x4 + x2\,(x3 + x4)) - 2\,x0\,(x3\,x4 + x2\,(x3 + x4) + x1\,(x2 + x3 + x4))}{(x0 - x1)\,(x0 - x2)\,(x0 - x3)\,(x0 - x4)},$

$b \to -\frac{(x0 - x2)\,(x0 - x3)\,(x0 - x4)}{(x0 - x1)\,(x1 - x2)\,(x1 - x3)\,(x1 - x4)},\ c \to -\frac{(x0 - x1)\,(x0 - x3)\,(x0 - x4)}{(x0 - x2)\,(-x1 + x2)\,(x2 - x3)\,(x2 - x4)},$

$d \to -\frac{(x0 - x1)\,(x0 - x2)\,(x0 - x4)}{(x0 - x3)\,(-x1 + x3)\,(-x2 + x3)\,(x3 - x4)},\ e \to -\frac{(x0 - x1)\,(x0 - x2)\,(x0 - x3)}{(x0 - x4)\,(-x1 + x4)\,(-x2 + x4)\,(-x3 + x4)}\Big\}\Big\}$

We now compute the symmetric five-point formula.

In[704]:=
```
symmcase = {x1 → x0 - 2 h, x2 → x0 - h, x3 → x0 + h, x4 → x0 + 2 h}
```

Out[704]= $\{x1 \to -2\,h + x0,\ x2 \to -h + x0,\ x3 \to h + x0,\ x4 \to 2\,h + x0\}$

In[705]:=
```
symmrule  = a f[x0] + b f[x1] + c f[x2] + d f[x3] + e f[x4] /. pt5[[1]] /. symmcase // Simplify
```

Out[705]=
$$\frac{f[-2\,h + x0] - 8\,f[-h + x0] + 8\,f[h + x0] - f[2\,h + x0]}{12\,h}$$

In[706]:=
```
Clear[fdif]
fdif[x0_, h_] = symmrule /. {f → ftest};
```

In[708]:=
```
x0 = 1; truth = ftest'[1];
TableForm[Join[{headings}, Results]]
```

Out[709]//TableForm=

epsilon	inf. prec.	error	finite prec.	error
0.01	1477.34	-1.89	1477.34	-1.89
0.001	-1657.51	3.09×10^{-4}	-1657.51	3.09×10^{-4}
0.0001	-1657.	2.96×10^{-8}	-1657.	2.94×10^{-8}
0.00001	-1657.	2.96×10^{-12}	-1657.	-6.45×10^{-10}
$1. \times 10^{-6}$	-1657.	4.44×10^{-16}	-1657.	1.68×10^{-8}
$1. \times 10^{-7}$	-1657.	0.	-1657.	-9.45×10^{-9}
$1. \times 10^{-8}$	-1657.	0.	-1657.	2.65×10^{-7}
$1. \times 10^{-9}$	-1657.	0.	-1656.96	-2.31×10^{-5}
$1. \times 10^{-10}$	-1657.	0.	-1657.03	1.6×10^{-5}
$1. \times 10^{-11}$	-1657.	0.	-1649.22	-4.7×10^{-3}
$1. \times 10^{-12}$	-1657.	0.	-1627.24	-1.8×10^{-2}
$1. \times 10^{-13}$	-1657.	0.	-1655.28	-1.04×10^{-3}
$1. \times 10^{-14}$	-1657.	0.	742.754	-1.45
$1. \times 10^{-15}$	-1657.	0.	-23 495.3	1.32×10^{1}

In[710]:=
```
error = ϵ / h + ErrorDv /. pt5[[1]] /. symmcase // Simplify
```

Out[710]=
$$\frac{\epsilon}{h} - \frac{1}{30} h^4 f^{(5)}[1]$$

In[711]:= `Solve[D[error, h] == 0, h]`

Out[711]= $\left\{\left\{h \to \frac{\left(-\frac{15}{2}\right)^{1/5} \epsilon^{1/5}}{f^{(5)}[1]^{1/5}}\right\}, \left\{h \to -\frac{\left(\frac{15}{2}\right)^{1/5} \epsilon^{1/5}}{f^{(5)}[1]^{1/5}}\right\}, \left\{h \to -\frac{(-1)^{2/5} \left(\frac{15}{2}\right)^{1/5} \epsilon^{1/5}}{f^{(5)}[1]^{1/5}}\right\}, \left\{h \to \frac{(-1)^{3/5} \left(\frac{15}{2}\right)^{1/5} \epsilon^{1/5}}{f^{(5)}[1]^{1/5}}\right\}, \left\{h \to -\frac{(-1)^{4/5} \left(\frac{15}{2}\right)^{1/5} \epsilon^{1/5}}{f^{(5)}[1]^{1/5}}\right\}\right\}$

In[712]:= $\text{maxerror} = -\frac{\epsilon}{h^2} - \frac{2}{15} h^3 M$

Out[712]= $-\frac{2 h^3 M}{15} - \frac{\epsilon}{h^2}$

In[713]:= `Solve[D[maxerror, h] == 0, h]`

Out[713]= $\left\{\left\{h \to -\frac{(-5)^{1/5} \epsilon^{1/5}}{M^{1/5}}\right\}, \left\{h \to \frac{5^{1/5} \epsilon^{1/5}}{M^{1/5}}\right\}, \left\{h \to \frac{(-1)^{2/5} 5^{1/5} \epsilon^{1/5}}{M^{1/5}}\right\}, \left\{h \to -\frac{(-1)^{3/5} 5^{1/5} \epsilon^{1/5}}{M^{1/5}}\right\}, \left\{h \to \frac{(-1)^{4/5} 5^{1/5} \epsilon^{1/5}}{M^{1/5}}\right\}\right\}$

Therefore, h should be about $\epsilon^{1/5}$, which is about 10^{-3} on double precision machines.

Five-Point Formula for the second derivative

Next we derive the seven-point formula for f''[x0].

In[714]:=
```
Tayf[x_] = Normal[Series[f[x],{x,x0,6}]]
```

Out[714]=
$$f[1] + (-1+x)\, f'[1] + \frac{1}{2}(-1+x)^2 f''[1] + \frac{1}{6}(-1+x)^3 f^{(3)}[1] + \frac{1}{24}(-1+x)^4 f^{(4)}[1] + \frac{1}{120}(-1+x)^5 f^{(5)}[1] + \frac{1}{720}(-1+x)^6 f^{(6)}[1]$$

In[715]:=
```
x2 = x0 - 2 h; x3 = x0 - h;
x4=x0 + h; x5 = x0 + 2 h;
```

In[717]:=
```
ErrorDv = ExpandAll[ a Tayf[x0] + c Tayf[x2] +
          d Tayf[x3] + e Tayf[x4] +
          af Tayf[x5] - f''[x0]];
```

The symmetric seven-point formula for f''[x0] is given by:

In[718]:=
```
pt72 =
Solve[{Coefficient[ErrorDv,f'[x0]]==0,Coefficient[ErrorDv,f[x0]]==0,Coefficient[ErrorDv,f''[x0]]==0,Coefficient[Err
orDv,f'''[x0]]==0,Coefficient[ErrorDv,f''''[x0]]==0},{a,c,d,e,af}]
```

Out[718]=
$$\left\{\left\{a \to -\frac{5}{2\,h^2},\ c \to -\frac{1}{12\,h^2},\ d \to \frac{4}{3\,h^2},\ e \to \frac{4}{3\,h^2},\ af \to -\frac{1}{12\,h^2}\right\}\right\}$$

In[719]:=
```
ErrorSymm = ErrorDv /. pt72
```

Out[719]=
$$\left\{-\frac{1}{90}\,h^4\,f^{(6)}[1]\right\}$$

Total error include the error in computing the values of f[x] at the points used; this error is proportional to machine epsilon, ϵ, which is about $10^{-16}$for double precision. Since evaluations of f[x] are divided by h, we arrive at the following formula for total error

In[720]:=
```
TotError = ϵ / h + ErrorSymm
```

Out[720]=
$$\left\{\frac{\epsilon}{h} - \frac{1}{90}\,h^4\,f^{(6)}[1]\right\}$$

To find the optimal h we find the minimum of TotError

In[721]:= `Solve[D[TotError, h] == 0, h]`

Out[721]=
$$\left\{\left\{h\to-\frac{(-3)^{2/5}\left(\frac{5}{2}\right)^{1/5}\epsilon^{1/5}}{f^{(6)}[1]^{1/5}}\right\},\left\{h\to\frac{\left(-\frac{5}{2}\right)^{1/5}3^{2/5}\epsilon^{1/5}}{f^{(6)}[1]^{1/5}}\right\},\left\{h\to-\frac{\left(\frac{5}{2}\right)^{1/5}3^{2/5}\epsilon^{1/5}}{f^{(6)}[1]^{1/5}}\right\},\left\{h\to\frac{(-1)^{3/5}\left(\frac{5}{2}\right)^{1/5}3^{2/5}\epsilon^{1/5}}{f^{(6)}[1]^{1/5}}\right\},\left\{h\to-\frac{(-1)^{4/5}\left(\frac{5}{2}\right)^{1/5}3^{2/5}\epsilon^{1/5}}{f^{(6)}[1]^{1/5}}\right\}\right\}$$

Therefore, h should be about $\epsilon^{1/5}$, which is about 10^{-3} on double precision machines.

Five-Point Formula for the third derivative

Next we derive the seven-point formula for f'''[x0].

In[722]:=
```
Tayf[x_] = Normal[Series[f[x],{x,x0,6}]]
```

Out[722]=
$$f[1]+(-1+x)\,f'[1]+\frac{1}{2}(-1+x)^2 f''[1]+\frac{1}{6}(-1+x)^3 f^{(3)}[1]+\frac{1}{24}(-1+x)^4 f^{(4)}[1]+\frac{1}{120}(-1+x)^5 f^{(5)}[1]+\frac{1}{720}(-1+x)^6 f^{(6)}[1]$$

In[723]:=
```
x2 = x0 - 2 h; x3 = x0 - h;
x4=x0 + h; x5 = x0 + 2 h;
```

In[725]:=
```
ErrorDv = ExpandAll[ a Tayf[x0] + c Tayf[x2] +
        d Tayf[x3] + e Tayf[x4] +
        af Tayf[x5] - f'''[x0]];
```

The symmetric seven-point formula for f''[x0] is given by:

In[726]:=
```
pt72 = Solve[{Coefficient[ErrorDv,f'[x0]]==0,
                Coefficient[ErrorDv,f[x0]]==0,
                Coefficient[ErrorDv,f''[x0]]==0,
                Coefficient[ErrorDv,f'''[x0]]==0,
                Coefficient[ErrorDv,f''''[x0]]==0},
             {a,c,d,e,af}]
```

Out[726]=
$$\left\{\left\{a\to 0,\ c\to -\frac{1}{2\,h^3},\ d\to \frac{1}{h^3},\ e\to -\frac{1}{h^3},\ af\to \frac{1}{2\,h^3}\right\}\right\}$$

In[727]:=
```
ErrorSymm = ErrorDv /. pt72
```

Out[727]=
$$\left\{\frac{1}{4} h^2 f^{(5)}[1]\right\}$$

Total error include the error in computing the values of f[x] at the points used; this error is proportional to machine epsilon, ϵ, which is about $10^{-16}$for double precision. Since evaluations of f[x] are divided by h, we arrive at the following formula for total error

In[728]:=
```
TotError = ϵ / h + ErrorSymm
```

Out[728]=
$$\left\{\frac{\epsilon}{h}+\frac{1}{4} h^2 f^{(5)}[1]\right\}$$

To find the optimal h we find the minimum of TotError

In[729]:=
```
Solve[D[TotError, h] == 0, h]
```

Out[729]= $\left\{\left\{h \to -\frac{(-2)^{1/3}\,\epsilon^{1/3}}{f^{(5)}[1]^{1/3}}\right\}, \left\{h \to \frac{2^{1/3}\,\epsilon^{1/3}}{f^{(5)}[1]^{1/3}}\right\}, \left\{h \to \frac{(-1)^{2/3}\,2^{1/3}\,\epsilon^{1/3}}{f^{(5)}[1]^{1/3}}\right\}\right\}$

Therefore, h should be about $\epsilon^{1/3}$, which is about 10^{-5} on double precision machines.

Seven-Point Formulas

Seven-Point Formula for the first derivative

Next we derive the seven-point formula for f'[x0].

In[730]:=
```
Tayf[x_] = Normal[Series[f[x],{x,x0,7}]]
```

In[731]:=
```
x1=x0 - 3 h; x2 = x0 - 2 h; x3 = x0 - h;
x4=x0 + h; x5 = x0 + 2 h; x6 = x0 + 3 h;
```

In[733]:=
```
ErrorDv = ExpandAll[ a Tayf[x0] + b Tayf[x1] + c Tayf[x2] +
        d Tayf[x3] + e Tayf[x4] +
        af Tayf[x5] + bf Tayf[x6]- f'[x0]];
```

The symmetric seven-point formula for f''[x0] is given by:

In[734]:=
```
pt72 = Solve[{Coefficient[ErrorDv,f'[x0]]==0,
                Coefficient[ErrorDv,f[x0]]==0,
                Coefficient[ErrorDv,f''[x0]]==0,
                Coefficient[ErrorDv,f'''[x0]]==0,
                Coefficient[ErrorDv,f''''[x0]]==0,
                Coefficient[ErrorDv,f'''''[x0]]==0,
                Coefficient[ErrorDv,f''''''[x0]]==0},
             {a,b,c,d,e,af,bf}]
```

Out[734]= $\left\{\left\{a \to 0,\ b \to -\frac{1}{60\,h},\ c \to \frac{3}{20\,h},\ d \to -\frac{3}{4\,h},\ e \to \frac{3}{4\,h},\ af \to -\frac{3}{20\,h},\ bf \to \frac{1}{60\,h}\right\}\right\}$

In[735]:=
```
ErrorSymm = ErrorDv /. pt72
```

Out[735]= $\left\{\frac{1}{140} h^6 f^{(7)}[1]\right\}$

Total error include the error in computing the values of f[x] at the points used; this error is proportional to machine epsilon, ϵ, which is about $10^{-16}$for double precision. Since evaluations of f[x] are divided by h, we arrive at the following formula for total error

In[736]:= `TotError = ϵ / h + ErrorSymm`

Out[736]= $\left\{\frac{\epsilon}{h}+\frac{1}{140}h^6 f^{(7)}[1]\right\}$

To find the optimal h we find the minimum of TotError

In[737]:= `Solve[D[TotError, h] == 0, h]`

Out[737]= $\left\{\left\{h\to-\frac{\left(-\frac{70}{3}\right)^{1/7}\epsilon^{1/7}}{f^{(7)}[1]^{1/7}}\right\},\left\{h\to\frac{\left(\frac{70}{3}\right)^{1/7}\epsilon^{1/7}}{f^{(7)}[1]^{1/7}}\right\},\left\{h\to\frac{(-1)^{2/7}\left(\frac{70}{3}\right)^{1/7}\epsilon^{1/7}}{f^{(7)}[1]^{1/7}}\right\},\left\{h\to-\frac{(-1)^{3/7}\left(\frac{70}{3}\right)^{1/7}\epsilon^{1/7}}{f^{(7)}[1]^{1/7}}\right\},\left\{h\to\frac{(-1)^{4/7}\left(\frac{70}{3}\right)^{1/7}\epsilon^{1/7}}{f^{(7)}[1]^{1/7}}\right\},\left\{h\to-\frac{(-1)^{5/7}\left(\frac{70}{3}\right)^{1/7}\epsilon^{1/7}}{f^{(7)}[1]^{1/7}}\right\},\left\{h\to\frac{(-1)^{6/7}\left(\frac{70}{3}\right)^{1/7}\epsilon^{1/7}}{f^{(7)}[1]^{1/7}}\right\}\right\}$

Therefore, h should be about $\epsilon^{1/7}$, which is about $10^{-2.2}$ on double precision machines.

Seven-Point Formula for the second derivative

Next we derive the seven-point formula for f''[x0].

In[738]:=
```
Tayf[x_] = Normal[Series[f[x],{x,x0,8}]]
```

Out[738]=
$$f[1] + (-1+x)\, f'[1] + \frac{1}{2}(-1+x)^2 f''[1] + \frac{1}{6}(-1+x)^3 f^{(3)}[1] + \frac{1}{24}(-1+x)^4 f^{(4)}[1] + \frac{1}{120}(-1+x)^5 f^{(5)}[1] + \frac{1}{720}(-1+x)^6 f^{(6)}[1] + \frac{(-1+x)^7 f^{(7)}[1]}{5040} + \frac{(-1+x)^8 f^{(8)}[1]}{40\,320}$$

In[739]:=
```
x1=x0 - 3 h; x2 = x0 - 2 h; x3 = x0 - h;
x4=x0 + h; x5 = x0 + 2 h; x6 = x0 + 3 h;
```

In[741]:=
```
ErrorDv = ExpandAll[ a Tayf[x0] + b Tayf[x1] + c Tayf[x2] +
        d Tayf[x3] + e Tayf[x4] +
        af Tayf[x5] + bf Tayf[x6]- f''[x0]];
```

The symmetric seven-point formula for f''[x0] is given by:

In[742]:=
```
pt72 = Solve[{Coefficient[ErrorDv,f'[x0]]==0,
                Coefficient[ErrorDv,f[x0]]==0,
                Coefficient[ErrorDv,f''[x0]]==0,
                Coefficient[ErrorDv,f'''[x0]]==0,
                Coefficient[ErrorDv,f''''[x0]]==0,
                Coefficient[ErrorDv,f'''''[x0]]==0,
                Coefficient[ErrorDv,f''''''[x0]]==0},
             {a,b,c,d,e,af,bf}]
```

Out[742]=
$$\left\{\left\{a \to -\frac{49}{18\,h^2},\ b \to \frac{1}{90\,h^2},\ c \to -\frac{3}{20\,h^2},\ d \to \frac{3}{2\,h^2},\ e \to \frac{3}{2\,h^2},\ af \to -\frac{3}{20\,h^2},\ bf \to \frac{1}{90\,h^2}\right\}\right\}$$

In[743]:=
```
ErrorSymm = ErrorDv /. pt72
```

Out[743]=
$$\left\{\frac{1}{560}\, h^6 f^{(8)}[1]\right\}$$

Total error include the error in computing the values of f[x] at the points used; this error is proportional to machine epsilon, ϵ, which is about $10^{-16}$for double precision. Since evaluations of f[x] are divided by h, we arrive at the following formula for total error

In[744]:= **TotError = ϵ / h + ErrorSymm**

Out[744]= $\left\{\frac{\epsilon}{h} + \frac{1}{560} h^6 f^{(8)}[1]\right\}$

To find the optimal h we find the minimum of TotError

In[745]:= **Solve[D[TotError, h] == 0, h]**

Out[745]= $\left\{\left\{h \to -\frac{\left(-\frac{35}{3}\right)^{1/7} 2^{3/7} \epsilon^{1/7}}{f^{(8)}[1]^{1/7}}\right\}, \left\{h \to -\frac{(-2)^{3/7} \left(\frac{35}{3}\right)^{1/7} \epsilon^{1/7}}{f^{(8)}[1]^{1/7}}\right\}, \left\{h \to \frac{2^{3/7} \left(\frac{35}{3}\right)^{1/7} \epsilon^{1/7}}{f^{(8)}[1]^{1/7}}\right\}, \left\{h \to \frac{(-1)^{2/7} 2^{3/7} \left(\frac{35}{3}\right)^{1/7} \epsilon^{1/7}}{f^{(8)}[1]^{1/7}}\right\}, \left\{h \to \frac{(-1)^{4/7} 2^{3/7} \left(\frac{35}{3}\right)^{1/7} \epsilon^{1/7}}{f^{(8)}[1]^{1/7}}\right\}, \left\{h \to -\frac{(-1)^{5/7} 2^{3/7} \left(\frac{35}{3}\right)^{1/7} \epsilon^{1/7}}{f^{(8)}[1]^{1/7}}\right\}, \left\{h \to \frac{(-1)^{6/7} 2^{3/7} \left(\frac{35}{3}\right)^{1/7} \epsilon^{1/7}}{f^{(8)}[1]^{1/7}}\right\}\right\}$

In[746]:= **% // N**

Out[746]= $\left\{\left\{h \to -\frac{(1.72244 + 0.829482\, i)\, \epsilon^{1/7}}{f^{(8)}[1.]^{1/7}}\right\}, \left\{h \to -\frac{(0.425407 + 1.86383\, i)\, \epsilon^{1/7}}{f^{(8)}[1.]^{1/7}}\right\}, \left\{h \to \frac{1.91176\, \epsilon^{1/7}}{f^{(8)}[1.]^{1/7}}\right\}, \left\{h \to \frac{(1.19196 + 1.49468\, i)\, \epsilon^{1/7}}{f^{(8)}[1.]^{1/7}}\right\}, \left\{h \to -\frac{(0.425407 - 1.86383\, i)\, \epsilon^{1/7}}{f^{(8)}[1.]^{1/7}}\right\}, \left\{h \to \frac{(1.19196 - 1.49468\, i)\, \epsilon^{1/7}}{f^{(8)}[1.]^{1/7}}\right\}, \left\{h \to -\frac{(1.72244 - 0.829482\, i)\, \epsilon^{1/7}}{f^{(8)}[1.]^{1/7}}\right\}\right\}$

Therefore, h should be about $\epsilon^{1/7}$, which is about $10^{-2.2}$ on double precision machines.

Seven-Point Formula for the third derivative

Next we derive the seven-point formula for f'''[x0].

In[747]:=
```
Tayf[x_] = Normal[Series[f[x],{x,x0,8}]]
```

Out[747]=

$$f[1] + (-1+x)\, f'[1] + \frac{1}{2}(-1+x)^2 f''[1] + \frac{1}{6}(-1+x)^3 f^{(3)}[1] + \frac{1}{24}(-1+x)^4 f^{(4)}[1] + \frac{1}{120}(-1+x)^5 f^{(5)}[1] + \frac{1}{720}(-1+x)^6 f^{(6)}[1] + \frac{(-1+x)^7 f^{(7)}[1]}{5040} + \frac{(-1+x)^8 f^{(8)}[1]}{40\,320}$$

In[748]:=
```
x1=x0 - 3 h; x2 = x0 - 2 h; x3 = x0 - h;
x4=x0 + h; x5 = x0 + 2 h; x6 = x0 + 3 h;
```

In[750]:=
```
ErrorDv = ExpandAll[ a Tayf[x0] + b Tayf[x1] + c Tayf[x2] +
        d Tayf[x3] + e Tayf[x4] +
        af Tayf[x5] + bf Tayf[x6]- f'''[x0]];
```

The symmetric seven-point formula for f''[x0] is given by:

In[751]:=
```
pt72 = Solve[{Coefficient[ErrorDv,f'[x0]]==0,
                Coefficient[ErrorDv,f[x0]]==0,
                Coefficient[ErrorDv,f''[x0]]==0,
                Coefficient[ErrorDv,f'''[x0]]==0,
                Coefficient[ErrorDv,f''''[x0]]==0,
                Coefficient[ErrorDv,f'''''[x0]]==0,
                Coefficient[ErrorDv,f''''''[x0]]==0},
             {a,b,c,d,e,af,bf}]
```

Out[751]=

$$\left\{\left\{a \to 0,\ b \to \frac{1}{8\,h^3},\ c \to -\frac{1}{h^3},\ d \to \frac{13}{8\,h^3},\ e \to -\frac{13}{8\,h^3},\ af \to \frac{1}{h^3},\ bf \to -\frac{1}{8\,h^3}\right\}\right\}$$

In[752]:=
```
ErrorSymm = ErrorDv /. pt72
```

Out[752]=

$$\left\{-\frac{7}{120} h^4 f^{(7)}[1]\right\}$$

Total error include the error in computing the values of f[x] at the points used; this error is proportional to machine epsilon, ϵ, which is about $10^{-16}$for double precision. Since evaluations of f[x] are divided by h, we arrive at the following formula for total error

In[753]:= `TotError = ϵ / h + ErrorSymm`

Out[753]= $\left\{\frac{\epsilon}{h} - \frac{7}{120} h^4 f^{(7)}[1]\right\}$

To find the optimal h we find the minimum of TotError

In[754]:= `Solve[D[TotError, h] == 0, h]`

Out[754]= $\left\{\left\{h \to \frac{\left(-\frac{30}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(7)}[1]^{1/5}}\right\}, \left\{h \to -\frac{\left(\frac{30}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(7)}[1]^{1/5}}\right\}, \left\{h \to -\frac{(-1)^{2/5} \left(\frac{30}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(7)}[1]^{1/5}}\right\}, \left\{h \to \frac{(-1)^{3/5} \left(\frac{30}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(7)}[1]^{1/5}}\right\}, \left\{h \to -\frac{(-1)^{4/5} \left(\frac{30}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(7)}[1]^{1/5}}\right\}\right\}$

Therefore, h should be about $\epsilon^{1/5}$, which is about 10^{-3} on double precision machines.

Seven-Point Formula for the fourth derivative

Next we derive the seven-point formula for f''''[x0].

In[755]:=
```
Tayf[x_] = Normal[Series[f[x],{x,x0,8}]]
```

In[756]:=
```
x1=x0 - 3 h; x2 = x0 - 2 h; x3 = x0 - h;
x4=x0 + h; x5 = x0 + 2 h; x6 = x0 + 3 h;
```

In[758]:=
```
ErrorDv = ExpandAll[ a Tayf[x0] + b Tayf[x1] + c Tayf[x2] +
        d Tayf[x3] + e Tayf[x4] +
        af Tayf[x5] + bf Tayf[x6]- f''''[x0]];
```

The symmetric seven-point formula for f''[x0] is given by:

In[759]:=
```
pt72 = Solve[{Coefficient[ErrorDv,f'[x0]]==0,
                Coefficient[ErrorDv,f[x0]]==0,
                Coefficient[ErrorDv,f''[x0]]==0,
                Coefficient[ErrorDv,f'''[x0]]==0,
                Coefficient[ErrorDv,f''''[x0]]==0,
                Coefficient[ErrorDv,f'''''[x0]]==0,
                Coefficient[ErrorDv,f''''''[x0]]==0},
             {a,b,c,d,e,af,bf}]
```

Out[759]= $\left\{\left\{a \to \frac{28}{3\,h^4},\ b \to -\frac{1}{6\,h^4},\ c \to \frac{2}{h^4},\ d \to -\frac{13}{2\,h^4},\ e \to -\frac{13}{2\,h^4},\ af \to \frac{2}{h^4},\ bf \to -\frac{1}{6\,h^4}\right\}\right\}$

In[760]:=
```
ErrorSymm = ErrorDv /. pt72
```

Out[760]= $\left\{-\frac{7}{240}\,h^4\,f^{(8)}[1]\right\}$

Total error include the error in computing the values of f[x] at the points used; this error is proportional to machine epsilon, ϵ, which is about $10^{-16}$for double precision. Since evaluations of f[x] are divided by h, we arrive at the following formula for total error

In[761]:=
```
TotError = ϵ / h + ErrorSymm
```

Out[761]= $\left\{\frac{\epsilon}{h} - \frac{7}{240}\,h^4\,f^{(8)}[1]\right\}$

To find the optimal h we find the minimum of TotError

In[762]:=
```
Solve[D[TotError, h] == 0, h]
```

Out[762]=
$$\left\{\left\{h \to \frac{\left(-\frac{15}{7}\right)^{1/5} 2^{2/5} \epsilon^{1/5}}{f^{(8)}[1]^{1/5}}\right\}, \left\{h \to -\frac{(-2)^{2/5} \left(\frac{15}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(8)}[1]^{1/5}}\right\}, \left\{h \to -\frac{2^{2/5} \left(\frac{15}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(8)}[1]^{1/5}}\right\}, \left\{h \to \frac{(-1)^{3/5} 2^{2/5} \left(\frac{15}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(8)}[1]^{1/5}}\right\}, \left\{h \to -\frac{(-1)^{4/5} 2^{2/5} \left(\frac{15}{7}\right)^{1/5} \epsilon^{1/5}}{f^{(8)}[1]^{1/5}}\right\}\right\}$$

Therefore, h should be about $\epsilon^{1/5}$, which is about 10^{-3} on double precision machines.

Nine-Point Formulas

Nine-Point Formula for second derivative

Next we derive the nine-point formula for f''[x0].

In[763]:=
```
Tayf[x_] = Normal[Series[f[x],{x,x0,9}]]
```

Out[763]=

$$f[1] + (-1+x)\, f'[1] + \frac{1}{2}(-1+x)^2 f''[1] + \frac{1}{6}(-1+x)^3 f^{(3)}[1] + \frac{1}{24}(-1+x)^4 f^{(4)}[1] + \frac{1}{120}(-1+x)^5 f^{(5)}[1] + \frac{1}{720}(-1+x)^6 f^{(6)}[1] + \frac{(-1+x)^7 f^{(7)}[1]}{5040} + \frac{(-1+x)^8 f^{(8)}[1]}{40\,320} + \frac{(-1+x)^9 f^{(9)}[1]}{362\,880}$$

In[764]:=
```
x1=x0 - 4 h; x2 = x0 - 3 h; x3 = x0 - 2 h;
x4 = x0 - h;
x5=x0 + h; x6 = x0 + 2 h; x7 = x0 + 3 h;
x8 = x0 + 4 h;
```

In[768]:=
```
ErrorDv = ExpandAll[ a Tayf[x0] + b Tayf[x1] + c Tayf[x2] +
        d Tayf[x3] + e Tayf[x4] +
        af Tayf[x5] + bf Tayf[x6] + cf Tayf[x7] +
        df Tayf[x8] - f'[x0]];
```

The symmetric seven-point formula for f''[x0] is given by:

In[769]:=
```
eqns = {Coefficient[ErrorDv, f'[x0]] == 0, Coefficient[ErrorDv, f[x0]] == 0,
   Coefficient[ErrorDv, f''[x0]] == 0, Coefficient[ErrorDv, f'''[x0]] == 0, Coefficient[ErrorDv, f''''[x0] ] == 0,
   Coefficient[ErrorDv, f'''''[x0]] == 0, Coefficient[ErrorDv, f''''''[x0]] == 0,
   Coefficient[ErrorDv, f'''''''[x0]] == 0, Coefficient[ErrorDv, f''''''''[x0]] == 0};
eqns // TableForm
vars = {a, b, c, d, e, af, bf, cf, df}
```

Out[770]//TableForm=

$$-1 + af\,h - 4\,b\,h + 2\,bf\,h - 3\,c\,h + 3\,cf\,h - 2\,d\,h + 4\,df\,h - e\,h == 0$$
$$a + af + b + bf + c + cf + d + df + e == 0$$
$$\frac{af\,h^2}{2} + 8\,b\,h^2 + 2\,bf\,h^2 + \frac{9\,c\,h^2}{2} + \frac{9\,cf\,h^2}{2} + 2\,d\,h^2 + 8\,df\,h^2 + \frac{e\,h^2}{2} == 0$$
$$\frac{af\,h^3}{6} - \frac{32\,b\,h^3}{3} + \frac{4\,bf\,h^3}{3} - \frac{9\,c\,h^3}{2} + \frac{9\,cf\,h^3}{2} - \frac{4\,d\,h^3}{3} + \frac{32\,df\,h^3}{3} - \frac{e\,h^3}{6} == 0$$
$$\frac{af\,h^4}{24} + \frac{32\,b\,h^4}{3} + \frac{2\,bf\,h^4}{3} + \frac{27\,c\,h^4}{8} + \frac{27\,cf\,h^4}{8} + \frac{2\,d\,h^4}{3} + \frac{32\,df\,h^4}{3} + \frac{e\,h^4}{24} == 0$$
$$\frac{af\,h^5}{120} - \frac{128\,b\,h^5}{15} + \frac{4\,bf\,h^5}{15} - \frac{81\,c\,h^5}{40} + \frac{81\,cf\,h^5}{40} - \frac{4\,d\,h^5}{15} + \frac{128\,df\,h^5}{15} - \frac{e\,h^5}{120} == 0$$
$$\frac{af\,h^6}{720} + \frac{256\,b\,h^6}{45} + \frac{4\,bf\,h^6}{45} + \frac{81\,c\,h^6}{80} + \frac{81\,cf\,h^6}{80} + \frac{4\,d\,h^6}{45} + \frac{256\,df\,h^6}{45} + \frac{e\,h^6}{720} == 0$$
$$\frac{af\,h^7}{5040} - \frac{1024\,b\,h^7}{315} + \frac{8\,bf\,h^7}{315} - \frac{243\,c\,h^7}{560} + \frac{243\,cf\,h^7}{560} - \frac{8\,d\,h^7}{315} + \frac{1024\,df\,h^7}{315} - \frac{e\,h^7}{5040} == 0$$
$$\frac{af\,h^8}{40\,320} + \frac{512\,b\,h^8}{315} + \frac{2\,bf\,h^8}{315} + \frac{729\,c\,h^8}{4480} + \frac{729\,cf\,h^8}{4480} + \frac{2\,d\,h^8}{315} + \frac{512\,df\,h^8}{315} + \frac{e\,h^8}{40\,320} == 0$$

Out[771]= {a, b, c, d, e, af, bf, cf, df}

In[772]:=
```
pt72 = Solve[eqns, vars]
```

Out[772]=
$$\left\{\left\{a \to 0,\ b \to \frac{1}{280\,h},\ c \to -\frac{4}{105\,h},\ d \to \frac{1}{5\,h},\ e \to -\frac{4}{5\,h},\ af \to \frac{4}{5\,h},\ bf \to -\frac{1}{5\,h},\ cf \to \frac{4}{105\,h},\ df \to -\frac{1}{280\,h}\right\}\right\}$$

In[773]:=
```
ErrorSymm = ErrorDv /. pt72
```

Out[773]=
$$\left\{-\frac{1}{630}\,h^8\,f^{(9)}[1]\right\}$$

Total error include the error in computing the values of f[x] at the points used; this error is proportional to machine epsilon, ϵ, which is about $10^{-16}$for double precision. Since evaluations of f[x] are divided by h, we arrive at the following formula for total error

In[774]:= **TotError = ϵ / h + ErrorSymm**

Out[774]= $\left\{\frac{\epsilon}{h} - \frac{1}{630} h^8 f^{(9)}[1]\right\}$

To find the optimal h we find the minimum of TotError

In[779]:= **Solve[D[TotError, h] == 0, h]**

Out[779]= $\left\{\left\{h \to \frac{(-35)^{1/9}\, 3^{2/9}\, \epsilon^{1/9}}{2^{2/9}\, f^{(9)}[1]^{1/9}}\right\}, \left\{h \to -\frac{\left(\frac{3}{2}\right)^{2/9} 35^{1/9}\, \epsilon^{1/9}}{f^{(9)}[1]^{1/9}}\right\}, \left\{h \to \frac{(-1)^{1/3}\left(\frac{3}{2}\right)^{2/9} 35^{1/9}\, \epsilon^{1/9}}{f^{(9)}[1]^{1/9}}\right\},\right.$
$\left\{h \to \frac{(-1)^{5/9}\left(\frac{3}{2}\right)^{2/9} 35^{1/9}\, \epsilon^{1/9}}{f^{(9)}[1]^{1/9}}\right\}, \left\{h \to \frac{(-1)^{7/9}\left(\frac{3}{2}\right)^{2/9} 35^{1/9}\, \epsilon^{1/9}}{f^{(9)}[1]^{1/9}}\right\}, \left\{h \to -\frac{(-3)^{2/9}\, 35^{1/9}\, \epsilon^{1/9}}{2^{2/9}\, f^{(9)}[1]^{1/9}}\right\},$
$\left.\left\{h \to -\frac{(-1)^{4/9}\, 3^{2/9} \times 35^{1/9}\, \epsilon^{1/9}}{2^{2/9}\, f^{(9)}[1]^{1/9}}\right\}, \left\{h \to -\frac{(-1)^{2/3}\, 3^{2/9} \times 35^{1/9}\, \epsilon^{1/9}}{2^{2/9}\, f^{(9)}[1]^{1/9}}\right\}, \left\{h \to -\frac{(-1)^{8/9}\, 3^{2/9} \times 35^{1/9}\, \epsilon^{1/9}}{2^{2/9}\, f^{(9)}[1]^{1/9}}\right\}\right\}$

Therefore, h should be about $\epsilon^{1/9}$, which is about 10^{-2} on double precision machines.

In[780]:= **% // N**

Out[780]= $\left\{\left\{h \to \frac{(1.52644 + 0.555579\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}, \left\{h \to -\frac{1.62441\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}, \left\{h \to \frac{(0.812203 + 1.40678\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\},\right.$
$\left\{h \to -\frac{(0.282075 - 1.59973\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}, \left\{h \to -\frac{(1.24437 - 1.04415\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}, \left\{h \to -\frac{(1.24437 + 1.04415\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\},$
$\left.\left\{h \to -\frac{(0.282075 + 1.59973\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}, \left\{h \to \frac{(0.812203 - 1.40678\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}, \left\{h \to \frac{(1.52644 - 0.555579\, i)\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}\right\}$

In[781]:= **%[[2]]**

Out[781]= $\left\{h \to -\frac{1.62441\, \epsilon^{1/9}}{f^{(9)}[1.]^{1/9}}\right\}$